Modeling Uncertain Relationships

10

10.1 BASE MODEL, FOUR INPUTS

Price is fixed. The three uncontrollable inputs are independent.

Figure 10.1 Four Inputs Influence Chart



Figure 10.2 Four Inputs Display

	A	B
1	**Controllable Input**	
2	Price	$29
3	**Uncontrollable Inputs**	
4	Fixed Costs	$12,000
5	Units Sold	700
6	Unit Variable Cost	$8
7	**Output Variable**	
8	Net Cash Flow	$2,700

Figure 10.3 Four Inputs Formulas

	A	B
1	**Controllable Input**	
2	Price	29
3	**Uncontrollable Inputs**	
4	Fixed Costs	12000
5	Units Sold	700
6	Unit Variable Cost	8
7	**Output Variable**	
8	Net Cash Flow	=(B2-B6)*B5-B4

10.2 THREE INPUTS

Price is variable. Units sold depends on price. The two cost inputs are independent.

Figure 10.4 Three Inputs Influence Chart



Figure 10.5 Three Inputs Display

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	$29		$29	700
3	**Uncontrollable Inputs**			$39	550
4	Fixed Costs	$12,000		$49	400
5	Unit Variable Cost	$8		$59	250
6	**Intermediate Variable**				
7	Units Sold	700		Slope	-15
8	**Output Variable**			Intercept	1135
9	Net Cash Flow	$2,700			

Figure 10.6 Three Inputs Formulas

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	29		29	700
3	**Uncontrollable Inputs**			39	550
4	Fixed Costs	12000		49	400
5	Unit Variable Cost	8		59	250
6	**Intermediate Variable**				
7	Units Sold	=E8+E7*B2		Slope	=SLOPE(E2:E5,D2:D5)
8	**Output Variable**			Intercept	=INTERCEPT(E2:E5,D2:D5)
9	Net Cash Flow	=(B2-B5)*B7-B4			

10.3 TWO INPUTS

Price is variable. Units sold depends on price. Unit variable cost depends on fixed costs.

Figure 10.7 Two Inputs Influence Chart



Figure 10.8 Two Inputs Display

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	$29		$29	700
3	**Uncontrollable Inputs**			$39	550
4	Fixed Costs	$12,000		$49	400
5	**Intermediate Variable**			$59	250
6	Unit Variable Cost	$8.00			
7	Units Sold	700		Slope	-15
8	**Output Variable**			Intercept	1135
9	Net Cash Flow	$2,700			
10					
11				Fixed Costs	Unit Variable Cost
12				$10,000	$11
13				$12,000	$8
14				$15,000	$6
15					
16				a	0.000000166667
17				b	-0.005166666667
18				c	46

Figure 10.9 Two Inputs Formulas

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	29		29	700
3	**Uncontrollable Inputs**			39	550
4	Fixed Costs	12000		49	400
5	**Intermediate Variable**			59	250
6	Unit Variable Cost	=E16*B4^2+E17*B4+E18			
7	Units Sold	=E8+E7*B2		Slope	=SLOPE(E2:E5,D2:D5)
8	**Output Variable**			Intercept	=INTERCEPT(E2:E5,D2:D5)
9	Net Cash Flow	=(B2-B6)*B7-B4			
10					
11				Fixed Costs	Unit Variable Cost
12				10000	11
13				12000	8
14				15000	6
15					
16				a	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
17				b	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
18				c	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))

10.4 FOUR INPUTS WITH THREE UNCERTAINTIES

Price is variable. Units sold depends on price. Unit variable cost depends on fixed costs. Fixed costs, units sold, and unit variable cost are uncertain.

Figure 10.10 Three Uncertainties Influence Chart



Figure 10.11 Three Uncertainties Display

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	$29		$29	700
3	**Uncontrollable Inputs**			$39	550
4	Fixed Costs	$12,000		$49	400
5	Units Sold Uncertainty	10		$59	250
6	Unit Variable Cost Uncertainty	$0.10			
7	**Intermediate Variable**			Slope	-15
8	Units Sold Median	700		Intercept	1135
9	Units Sold	710			
10	Unit Variable Cost Median	$8.00			
11	Unit Variable Cost	$8.10		Fixed Costs	Unit Variable Cost
12	**Output Variable**			$10,000	$11
13	Net Cash Flow	$2,839		$12,000	$8
14				$15,000	$6
15					
16				a	0.000000166667
17				b	-0.005166666667
18				c	46

Figure 10.12 Three Uncertainties Formulas

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	29		29	700
3	**Uncontrollable Inputs**			39	550
4	Fixed Costs	12000		49	400
5	Units Sold Uncertainty	10		59	250
6	Unit Variable Cost Uncertainty	0.1			
7	**Intermediate Variable**			Slope	=SLOPE(E2:E5,D2:D5)
8	Units Sold Median	=E8+E7*B2		Intercept	=INTERCEPT(E2:E5,D2:D5)
9	Units Sold	=B8+B5			
10	Unit Variable Cost Median	=E16*B4^2+E17*B4+E18			
11	Unit Variable Cost	=B10+B6		Fixed Costs	Unit Variable Cost
12	**Output Variable**			10000	11
13	Net Cash Flow	=(B2-B11)*B9-B4		12000	8
14				15000	6
15					
16				a	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
17				b	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
18				c	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))

10.5 INTERMEDIATE DETAILS

Price is variable. Units sold depends on price. Unit variable cost depends on fixed costs.

Fixed costs, units sold, and unit variable cost are uncertain.

Include revenue, total variable cost, and total costs as intermediate variables.

Figure 10.13 Intermediate Details Influence Chart



Figure 10.14 Intermediate Details Display

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	$29		$29	700
3	**Uncontrollable Inputs**			$39	550
4	Fixed Costs	$12,000		$49	400
5	Units Sold Uncertainty	10		$59	250
6	Unit Variable Cost Uncertainty	$0.10			
7	**Intermediate Variable**			Slope	-15
8	Units Sold Median	700		Intercept	1135
9	Units Sold	710			
10	Revenue	$20,590			
11	Unit Variable Cost Median	$8.00		Fixed Costs	Unit Variable Cost
12	Unit Variable Cost	$8.10		$10,000	$11
13	Total Variable Cost	$5,751		$12,000	$8
14	Total Costs	$17,751		$15,000	$6
15	**Output Variable**				
16	Net Cash Flow	$2,839		a	0.000000166667
17				b	-0.005166666667
18				c	46

Figure 10.15 Intermediate Details Formulas

	A	B	C	D	E
1	**Controllable Input**			Price	Units Sold
2	Price	29		29	700
3	**Uncontrollable Inputs**			39	550
4	Fixed Costs	12000		49	400
5	Units Sold Uncertainty	10		59	250
6	Unit Variable Cost Uncertainty	0.1			
7	**Intermediate Variable**			Slope	=SLOPE(E2:E5,D2:D5)
8	Units Sold Median	=E8+E7*B2		Intercept	=INTERCEPT(E2:E5,D2:D5)
9	Units Sold	=B8+B5			
10	Revenue	=B9*B2			
11	Unit Variable Cost Median	=E16*B4^2+E17*B4+E18		Fixed Costs	Unit Variable Cost
12	Unit Variable Cost	=B11+B6		10000	11
13	Total Variable Cost	=B12*B9		12000	8
14	Total Costs	=B4+B13		15000	6
15	**Output Variable**				
16	Net Cash Flow	=B10-B14		a	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
17				b	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))
18				c	=TRANSPOSE(LINEST(E12:E14,D12:D14^{1,2}))

10.6 BIVARIATE SIMULATION MODELS

This section is a paper entitled "Spreadsheet Simulation Models for Business Decisions with Uncertain Relationships," presented at the Southwest Decision Sciences Meeting in March 2007.

Introduction

A decision maker is analyzing the uncertainty about cash flows associated with introducing a new product. I examine the situation of a price taker who does not have control over the future price and quantity demanded. I describe three methods for developing risk profiles: simulation by specifying directly the correlation of price and quantity, simulation by specifying uncertain quantity dependent on uncertain price, and a discrete approximation of quantity dependent on price. I illustrate these situations with influence diagrams, spreadsheet model displays, and charts, and I compare the risk profiles of the three methods.

An independent software vendor (ISV) must decide whether to introduce a new product. If he decides to develop the software, he plans to charge the prevailing market price for similar products. The ISV uses net cash flow as his performance measure, and he wants to describe the uncertain net cash flow so that he can compare this project with others he is considering. He uses the general approach of decomposition described by Ravinder (2000) to develop a model for determining how net cash flow depends on other factors that are easier to assess. Revenue depends on price and quantity, but he is uncertain about what the price will be and the number of units he will be able to sell during the product's lifetime. He estimates that with a market price of $30, he will sell 900 units; if the price is $60, he will sell only 450 units. His estimate of the demand curve is shown in Figure 1.



Figure 1: Quantity Dependent on Price

Model Using Correlation of Quantity and Price

Clemen (2000) used six methods for assessing dependence and concluded that direct specification of the correlation coefficient was generally superior for the subjects in his experiments. Figure 2 is an influence diagram for the ISV's situation, and Figures 3 and 4 show the associated spreadsheet model.



Figure 2: Influence Diagram with Correlated Price and Quantity

Although there is also some uncertainty about unit variable cost and fixed cost, I will keep those input assumptions at their best-guess values and focus on the uncertainty about price and quantity. The intermediate variables are not necessary, but they can be helpful when you build and debug the spreadsheet model.

	A	B	C	D	E	F	G	H	I
1	Uncontrollable Inputs			Mean	StDev	Correl(P,Q)		Price	Quantity
2		Price	$45.00	$45.00	$7.50	-0.80		$30	900
3		Quantity	675	675	112.5			$40	750
4		Unit Var. Cost	$8.00					$50	600
5		Fixed Cost	$4,000					$60	450
6	Intermediate Variables								
7		Revenue	$30,375						Assessments
8		Total Var. Cost	$5,400					Intercept	1350
9		Total Cost	$9,400					Slope	-15
10	Performance Measure							Correlation	-0.80
11		Net Cash Flow	$20,975						

Figure 3: Spreadsheet Display for Correlation Model

The ISV has decided to use truncated normal distributions to describe his uncertainty about price and quantity. Each distribution ranges plus-and-minus two standard deviations from the mean and encompasses the range of prices and quantities that the ISV judges to be possible. The dependency is expressed with a preliminary estimate of R = –0.80. The slope and intercept are not used in this model, but they may be useful for checking whether the correlation agrees with the original assessments of the demand curve.



Figure 4: Truncated Normal Marginal Distributions for Price and Quantity

For simulation, the spreadsheet model uses a truncated bivariate normal random number generator function from RiskSim, a Monte Carlo simulation add-in for Excel. Its arguments are RandTruncBiVarNormal(Mean1,StDev1,Mean2,StDev2,Correl12,Min1,Max1,Min2,Max2), as shown in Figure 5.

	C	D	E	F	G	H	I
1		Mean	StDev	Correl(P,Q)		Price	Quantity
2	=RANDTruncBIVARNORMAL(D2,E2,D3,E3,F2,H2,H5,I5,I2)	45	7.5	-0.8		30	900
3	=RANDTruncBIVARNORMAL(D2,E2,D3,E3,F2,H2,H5,I5,I2)	675	112.5			40	750
4	8					50	600
5	4000					60	450
6							
7	=C2*C3						Assessments
8	=C4*C3					Intercept	=INTERCEPT(I2:I5,H2:H5)
9	=C5+C8					Slope	=SLOPE(I2:I5,H2:H5)
10						Correlation	=F2
11	=C7-C9						

Figure 5: Spreadsheet Formulas for Correlation Model

When R (= Correl12) is set to 0.00, a simulation of 100 trials yields the scatter plot shown in Figure 6. The pattern is completely random and consistent with the normally-distributed marginal distributions. The estimated demand curve is shown as a dashed diagonal line in these scatter plots, and the best-fit regression line is solid. For R = 0.00, the best-fit line is horizontal.

When we introduce some negative correlation using R = –0.50, the best-fit line becomes somewhat diagonal, but it does not agree with the estimated demand curve. To be consistent, make the best-fit line of the scatter plot approximately equal to the estimated demand curve.



Figure 6: Correlation Plots for R = 0.00 and R = –0.50

The correlation chart for R = –0.50 using the textbook edition of Crystal Ball is shown in Figure 7. The marginal normal distributions are not truncated, and Crystal Ball shows many more points, but the scatter is similar to Figure 6. Crystal Ball's dialog box shows only the diagonal line, but the best-fit line may also be useful for this kind of subjective assessment.



Figure 7: Crystal Ball Correlation Plot for R = –0.50

As the value of R approached –1.00, the best-fit line tends to agree with the diagonal line, as shown in Figure 8.



Figure 8: Correlation Plots for R = –0.80 and R = –0.90

Correlation plots like those in Figures 6 and 8 should be helpful for assessing dependence for uncertain relationships. Referring to the plot for R = –0.80 in Figure 8, the decision analyst could verify the assessment by asking the decision maker questions like: "If the price is $40, do you think you will sell between 620 and 820 units?" This kind of question leads to another natural way to think about the uncertainty, i.e., an uncertain quantity dependent on an uncertain price.

Model Using Quantity Dependent on Price

To aid the assessment of dependency using conditional probabilities, I suggest using the demand curve as an intermediate step. First, a marginal probability distribution is assigned to price.

Second, for a given price, I use the demand curve shown in Figure 1 to determine the median quantity. Third, I add an uncertainty adjustment to the median to determine actual demand, as shown in the influence diagram of Figure 9.



Figure 9: Influence Diagram with Quantity Dependent on Price

Figure 10 displays a simulation trial where the random price is $45.07, the median quantity (from Figure 1) is 674, the random quantity adjustment is –75, and the resulting random quantity is 599.

	A	B	C	D	E	F	G	H
1	Uncontrollable Inputs			Mean	StDev		Price	Quantity
2		Price	$45.07	$45.00	$7.50		$30	900
3		Quantity Uncertainty	-75	0	75		$40	750
4		Unit Var. Cost	$8				$50	600
5		Fixed Cost	$4,000				$60	450
6	Intermediate Variables							
7		Median Quantity	674				Intercept	1350
8		Quantity	599				Slope	-15
9		Revenue	$26,997					
10		Total Var. Cost	$5,392					
11		Total Cost	$9,392					
12	Performance Measure							
13		Net Cash Flow	$17,605					

Figure 10: Spreadsheet Display for Dependency Model

This spreadsheet model uses RiskSim's truncated normal random number generator function, whose arguments are RandTruncNormal(Mean1,StDev1,Mean2,StDev2,Correl12). The formulas for the model are shown in Figure 11.

	A	B	C	D	E	F	G	H
1	Uncontrollable Inputs			Mean	StDev		Price	Quantity
2		Price	=ROUND(randtruncnormal(D2,E2,G2,G5),2)	45	7.5		30	900
3		Quantity Uncertainty	=ROUND(randtruncnormal(D3,E3,-2*E3,2*E3),0)	0	75		40	750
4		Unit Var. Cost	8				50	600
5		Fixed Cost	4000				60	450
6	Intermediate Variables							
7		Median Quantity	=ROUND(H7+H8*C2,0)				Intercept	=INTERCEPT(H2:H5,G2:G5)
8		Quantity	=C7+C3				Slope	=SLOPE(H2:H5,G2:G5)
9		Revenue	=C2*C8					
10		Total Var. Cost	=C4*C7					
11		Total Cost	=C5+C10					
12	Performance Measure							
13		Net Cash Flow	=C9-C11					

Figure 11: Spreadsheet Formulas for Dependency Model

A simulation of 100 trials yields the scatter plot shown in Figure 12, with R = –0.80. Other simulations with this small sample size will produce somewhat different plots and values of R. There is close agreement with the R = –0.80 plot using the correlation model in Figure 8. An advantage of the dependency model is that, on the average, the best-fit line will be equal to the estimated demand curve.



Figure 12: Correlation Plot for Dependency Model

Model Using Discrete Approximations

A third approach to modeling the dependency is to use discrete probability for the marginal price distribution and discrete conditional probability for the quantity. To be somewhat consistent with the previous methods, I choose to use discrete approximations of the normal distributions. McNamee (1990) suggests the 10-50-90 shortcut method where those fractiles of the continuous distribution are assigned discrete probabilities 0.25, 0.50, and 0.25, respectively.

The 10% and 90% fractiles of a normal distribution are 1.645 standard deviations from the mean. For normally distributed price with mean \$45 and standard deviation \$7.50, the extreme fractiles are approximately \$33 and \$57, as shown in Figure 13. The corresponding conditional medians for quantity from the demand curve are 900, 675, and 450. For the quantity adjustment, the mean is 0 and the standard deviation is 75, so the adjustments are 1.645*75 = 123, yielding the quantity values shown in Figure 13. The nine endpoints of the probability tree describe nine combinations of price and quantity. The joint probability is calculated as Prob(P&Q) = Prob(Q|P)*Prob(P).



Figure 13: Probability Tree for Discrete Approximation

In Figure 14, the area of each bubble is proportional to the joint probability. The total of the bubble areas is probability 1.00. Thus, Figure 14 is somewhat similar to the correlation plots in Figures 6, 8, and 12. The correlation coefficient for discrete approximation (calculations not shown here) yields R = –0.88.



Figure 14: Bubble Chart for Discrete Approximation

Conclusions

I have described three methods for describing the dependency between price and quantity for the ISV price taker. Using the general methodology of modern decision analysis, a next step is to use these input assumptions to develop the payoff distribution, i.e., the risk profile. For each of the two spreadsheet models, I used RiskSim to perform a simulation of 1,000 trials. For the discrete approximation, I used a spreadsheet model like Figure 3 to compute net cash flow for each of the nine combinations of price and quantity, and the net cash flows shown in Figure 13 were sorted to develop the discrete cumulative distribution (not shown here). Figure 15 shows the cumulative probability distributions for the three methods. As expected, the simulation results for the three methods are quite similar.



Figure 15: Risk Profiles for the Three Methods

The best-guess estimate of price is $45, the best-guess estimate of quantity is 675, and the corresponding best-guess of net cash flow is $20,975. One insight is that the simulation results indicate there is a 40% chance of achieving that value. If the ISV wants to compare this project with other alternatives, the next steps are to check for outcome dominance and probabilistic dominance or explicitly assess attitude toward risk.

I have described three methods that should make it easier for MBA students and professional analysts to model dependency. For a more advanced example, McMullen (2005) models dependency among three variables by directly specifying correlations.

References

Clemen, R., Fischer, G., and Winkler, R. (2000). Assessing Dependence: Some Experimental Results. *Management Science 46*(8), 1100 – 1115.

McNamee, P., and Celona, J. (1990). Decision Analysis with Supertree, 2nd ed. San Francisco: The Scientific Press.

McMullen, P. (2005). Using Correlation Matrices and Optimization to Add Practical Functionality to Spreadsheet Simulation for MBA-Level Quantitative Analysis Courses. *Decision Sciences Journal of Innovative Education 3*(1), 161 – 168.

Ravinder, H., Kleinmuntz, D., and Dyer, J. (1988). The Reliability of Subjective Probabilities Obtained Through Decomposition. *Management Science 34*(2), 186 – 199.